
5 Numerische Methoden der praktischen Mathematik 

5.1 Einführung 
Zwischen Mathematik und Praxis (Technik, Physik, Ökonomie, ...) ist über die 
Jahrhunderte hinweg eine befruchtende Wechselwirkung zu beobachten: 
 
Mit den Rechenhilfsmitteln wurden zunächst nur formelmäβig lösbare Probleme 
mathematisch bearbeitet. Später, zur Zeit der mechanischen Rechenmaschinen, 
entwickelten die Mathematiker Näherungsverfahren, welche mittels sogenannten 
Rechenschemata gelöst wurden. Heute gestatten die modernen EDV-Anlagen neue 
numerische Lösungsverfahren, welche als Programme auf diesen Anlagen ausgeführt 
werden. 
 
Ausgangspunkt der Lösung praktischer Probleme mittels der Mathematik sind häufig 
durch Messungen ermittelte Werte. 
⇒ Fehlerbehaftete Eingabewerte, deren Fehlertoleranzen von der verwendeten 

Messtechnik abhängen. 
 
Numerische Verfahren sind oft Näherungsverfahren. 
⇒ Verfahrensfehler, welche sich bei der Berechnung fortpflanzen. 
 
Mathematische Hilfsmittel, früher Rechenschieber, heute EDV-Anlagen, können nicht 
unbegrenzt genau arbeiten. 
⇒ Rechenfehler, diese werden zwar durch die immer genauer arbeitenden Computer 

geringer, sind aber durch die Endlichkeit des Speichers und damit der 
Zahlendarstellung nicht zu beseitigen. 

 
Die Lösung des Problems ist folglich mit Eingabefehlern, Verfahrensfehlern und 
Rechenfehlern behaftet. Die Korrektheit einer Lösung wird durch 
Fehlerfortpflanzung beeinflusst: Fehlerbehaftete Eingabewerten werden mittels 
fehlerbehafteten Verfahren und ungenauer Rechnung bearbeitet. 
 
Die Numerik beschäftigt sich mit Verfahren zur Lösung mathematische Probleme, 
deren Fehlerfortpflanzung und Ergebnisauswertung. Einem numerische Verfahren 
liegt stets ein Algorithmus zugrunde, es enthält Kontrollen zur Fehlerfortpflanzung 
und entscheidet, in welchem Bereich die Lösungsmethode angewendet werden kann. 
Bei der Auswahl des Verfahrens sind Rechenzeit und Speicherökonomie stets zu 
beachten. Man beurteilt ein numerisches Verfahren streng nach ökonomischen 
Gesichtspunkten. Mathematisch elegante Lösungen sind häufig numerisch ungünstig. 
 
Prinzip: Ein verwendetes numerisches Verfahren muss die notwendige 
Genauigkeit gewähren, aber unnötigen Rechenaufwand vermeiden. 

5.2 Lösung linearer Gleichungssysteme 
Grundgedanke 
Durch Umformen der Koeffizientenmatrix in eine geeignete Form ( Dreiecks- oder 
Diagonalmatrix) werden Unbekannte eliminiert. 
Ausgangspunkt 
Lineares inhomogenes oder homogenes Gleichungssystem mit einer 
Koeffizientendeterminante ≠ 0. 
 
Das Eliminationsverfahren von Carl-Friedrich Gauβ (1777-1855) 
Durch Multiplikation und Addition von Gleichungen wird eine Koeffizientenmatrix in 
Dreiecksgestalt erzeugt, d.h. eine Gleichung mit einer Unbekannten, eine Gleichung 
mit zwei Unbekannten usw. usf. Danach wird aus der Gleichung mit einer 
Unbekannten diese berechnet, dann in die Gleichung mit zwei Unbekannten 
eingesetzt und die zweite Unbekannte berechnet usw. usf. bis alle Unbekannten 
aufgelöst sind. 
 
Betrachte drei lineare Gleichungen mit drei Unbekannten ,  und . 1x 2x 3x

Rechenschema für ein lineares Gleichungssystem mit 3 Unbekannten s. nächste Seite 
 
Beispiel 

Rechenschema: 

1 -2 1 0  x1 = 1
2 -5 2 -2 q = -2 

-2 4 -2 0  
2 -2 3 7 q = -2 

-2 4 -2 0  
0 -1 0 -2  x2 = 2
0 2 1 7 q =  2 
0 -2 0 -4  
0 0 1 3  x3 = 3

 
Verkürztes Rechenschema: 

1 -2 1 0  1
2 -5 2 -2 -2 
2 -2 3 7 -2 
0 -1 0 -2  2
0 2 1 7 2 
0 0 1 3  3

Rechenschema: 
 
 
 
Elimination von : 1x  11a  

21a  

11qa  

31a  

11qa  
Elimination von : 2x  0  

0  

0  

0  

  

  
  

  

  

  

  
+ 

+ 

+ 

 
 
Voraussetzung: , ,  ≠ 011a 22a ′ 33a ′′
 

Informatik/Numerik (2002) 
 

12a  

22a  

12qa  

32a  

12qa  

22a ′  

32a ′  

22aq ′  
0  

 

 

13a

23a

1qa

33a

1qa

23a ′

33a ′

2aq ′

33a ′′
Institut für Informatik 
Dr. Monika Meiler 

 1a    

11

3132121
1 a

xaxaa
x

−−
=  

 2a  

11

21

a
a

q −=

3  1qa  
 3a  

11

31

a
a

q −=  

3  1qa  
 2a ′  

22

3232
2 a

xaa
x

′
′−′

=  

 3a ′  
22

32

a
aq
′
′

−=  

3  2aq ′  

 3a ′′  

33

3
3 a

a
x

′′
′′

=  

  

   
  

  
  

 
*

  

  

  

 
*

*

5-3/27 



Universität Leipzig  Institut für Informatik 
Dr. Monika Meiler 

Gauss.bas 
REM Gauss-Algorithmus 
REM mit Unterprogrammtechnik und strukturierten Variablen 
REM M. Meiler 
REM 07.01.2002 
 
REM Hauptprogramm 
CLS : GOSUB eingabe: GOSUB gauss: GOSUB ausgabe: GOSUB ergebnis 
END 
 
REM Unterprogramme 
eingabe:                         'zeilenweises Einlesen der Matrix 
INPUT "Anzahl der Gleichungen: ", Laenge% 
DIM a(Laenge%, Laenge% + 1)    'Vereinbarung einer Matrix als Feld 
PRINT "Matrixeingabe" 
FOR i% = 1 TO Laenge% 
  FOR j% = 1 TO Laenge% + 1              '+1: rechte Seite des LGS 
    INPUT ; "  ", a(i%, j%) 
  NEXT j% 
  PRINT                                            'Zeilenvorschub 
NEXT i% 
RETURN 
 
gauss:            'Gauss-Algorithmus: Erzeugen der Dreiecksgestalt 
FOR k% = 1 TO Laenge% - 1 
  FOR i% = k% + 1 TO Laenge% 
    q = -a(i%, k%) / a(k%, k%) 
    ' PRINT "q="; q 
    FOR j% = k% TO Laenge% + 1 
      a(i%, j%) = a(i%, j%) + q * a(k%, j%) 
    NEXT j% 
    ' GOSUB ausgabe 
  NEXT i% 
NEXT k% 
RETURN 
 
ergebnis:                                    'Berechnen der Lösung 
DIM x(Laenge%)        'Vereinbarung eines Loesungsvektors als Feld 
FOR i% = Laenge% TO 1 STEP -1               'Berechnen der Loesung 
  x(i%) = a(i%, Laenge% + 1) 
  FOR j% = i% + 1 TO Laenge% 
    x(i%) = x(i%) - a(i%, j%) * x(j%) 
  NEXT j% 
  x(i%) = x(i%) / a(i%, i%) 
NEXT i% 
PRINT "Loesung: ";                            'Ausgabe der Loesung 
FOR i% = 1 TO Laenge% 
  PRINT x(i%); 
NEXT i% 
RETURN 
 
ausgabe:                         'zeilenweise Ausgabe einer Matrix 
PRINT "Matrixausgabe" 
FOR ii% = 1 TO Laenge% 
  FOR jj% = 1 TO Laenge% + 1 
    PRINT "  "; a(ii%, jj%); 
  NEXT jj% 
  PRINT 
NEXT ii% 
INPUT "Weiter mit ENTER ", e$ 
RETURN 

5.3 Numerische Integration 
Die numerische Integration beschäftigt sich mit der angenäherten zahlenmäβigen 
Berechnung bestimmter Intergrale. 
 

( )dxxfF
b

a∫=  

 
Grundgedanke 
Die Funktion  wird näherungsweise durch ein Interpolationspolynom , r 
Grad des Polynoms, dargestellt. 

( )xf ( )xPr

 

( ) ( )dxxPdxxfF
b

a r

b

a ∫∫ ≈=  

 
Ausgangspunkt 
Gegeben sind  äquidistante (gleicher Abstand) Stützstellen  mit 

,  und  (d.h. ) für  und ihre 
Funktionswerte . 

1+n

y =0

nxxx ,,, 10 L

1, −nLax =0 bxn = hxx ii +=+1

( ) fyxf =, 10

( )hixxi 101 ++=+

) ( )nn xfy =,L

,1,0=i
(x ,1

 
Prinzip 
Durch die Punkte (  werden Polynome gelegt, welche die Funktion angenähert 
darstellt. 

)ii yx ,

 
 

5.3.1 Trapezregel 
Integration durch ein lineares Interpolationspolynom ( ) 011 axaxP += : 

⇒ Je mehr Stützstellen, desto genauer wird die Annäherung an die eigentliche 
Funktion. 

Die Anwendung der Trapezflächenberechnung ( ) ( 11 2
1

++=+

ii

x

x

i yyhdxxPi

i

)∫  liefert dann 

( ) ( ) ( ) ( ) ( )3210322110 22
2222

2

0

yyyyhyyhyyhyyhdxxf
x

x
+++=+++++≈∫ . 

 
Allgemein 
[ ba. ] ] wird in n Teilintervalle  mit ,  zerlegt. In 
jedem dieser Teilintervalle wird die Funktion  durch ein lineares 
Interpolationspolynom angenähert. Das Integral über diese Polynome erhält man 
mittels der Flächenberechnung des entsprechenden Trapezes: 

[ ] [ ] [ nn xxxxxx .,.,. 12110 −L ax =0

( )x
bxn =

f

 

  Trapezregel:  ( ) ( )nn

b

a
yyyyyhdxxfF +++++≈= −∫ 1210 222

2
L   .  

 
Beispiel 

Zu berechnen sei das Flächenintegral ∫ +
1

0 1 x
dx  durch lineare Interpolation. Es sollen 

dabei sieben Stützstellen verwendet werden ⇒ n , a , b , 6= 0= 1=
6
1=−=

n
abh : 

 
i 0 1 2 3 4 5 6 
ix  0 6

1  3
1  2

1  3
2  6

5  1 

iy  1 7
6  4

3  3
2  5

3  11
6  2

1  
TR 1 0.85714 0.75 0.666667 0.6 0.54545 0.5 

 
 

Mathematische Lösung: ∫∫ =
+

2

1

1

0 1 t
dt

x
dx  mit 11 =⇒+=

dx
dtxt  

≈=−=−== 2ln02ln1ln2lnln
2

1
t  0.6931 (TR). 

 

Trapezregel: ≈





 ++++++≈

+∫ 2
1

11
12

5
6

3
4

2
3

7
121

12
1

1
1

0 x
dx  0.6949 (TR) 

 
 
Programm Trapez.bas : 0.6948762 (einfache Genauigkeit) 
 0.6948761666666 (doppelte Genauigkeit) 
 
Durch eine genauere Rechnung erhält man kein besseres Ergebnis. Die 
Verfahrensfehler kompensieren sich nicht durch die Erhöhung der Rechengenauigkeit. 

Trapez.bas 
REM Berechnung eines bestimmten Integrals 
REM mittels Trapezregel 
REM M. Meiler, 23.01.1998 
 
REM Hauptprogramm 
CLS : GOSUB eingabe: GOSUB trapez: GOSUB ausgabe 
END 
 
 
REM Eingabe der obere und unteren Grenze 
REM und der  Stuetzstellen 
eingabe: 
DO 
  INPUT "Untere Integrationsgrenze a= ", a! 
  INPUT "Obere Integrationsgrenze b(b>a)= ", b! 
LOOP UNTIL (b! - a!) > 0 
 
DO 
  INPUT "Anzahl der Stuetzstellen (s > 1): ", s% 
LOOP UNTIL s% >= 2 
 
n% = s% - 1 
h! = (b! - a!) / n% 
DIM y!(n%) 
 
FOR i% = 0 TO n% 
  LOCATE , 5 
  PRINT "x("; i%; ")="; a! + i% * h!; 
  LOCATE , 30 
  PRINT "y("; i%; ")="; 
  INPUT " ", y!(i%) 
NEXT i% 
RETURN 
 
 
REM Berechnung des Integrals mittels der Trapezregel 
trapez: 
y! = y!(0) + y!(n%) 
 
FOR j% = 1 TO n% - 1 
  y! = y! + 2 * y!(j%) 
NEXT j% 
 
y! = y! * h! / 2 
RETURN 
 
 
REM Ausgabe des Ergebnisses 
ausgabe: 
PRINT "Ergebnis: "; y! 
PRINT 
RETURN 

5.3.2 Simpsonregel 
Integration durch Interpolation mit Polynomen zweiten Grades (Parabeln) 

( ) 01
2

22 axaxaxP ++= . 
 

 

P2(x)

y2

a = x0

h

y1y0

x

f(x)

b = x2x1

y 

 
 
Gegeben sind drei Stützstellen, durch die eine Parabel zu legen ist. 
Die Berechnung der Koeffizienten der Parabel erfolgt mittels Gauβschem 
Eliminationsverfahren unter Berücksichtigung, dass die Stützstellen der Funktion 
Punkte der Parabel sind: 

h

( ) ( )( 102010 xxxxxxy −−∆+−∆+=  
 
Parabelgleichung:  ( ) ( ) ( )( 10201001

2
22 xxxxxxyaxaxaxP −−∆+−∆+=++=

mit 
h

yy 01
1

−=∆  und 2
102

2 2

h

Integralberechnung 
( ) ( ) ( )( )( )∫∫ −−∆+−∆+= 2

  Keplersche Fassregel:  ( ) ( )210 4
3

2

0

yyyhdxxfF
x

x
++≈= ∫   .  

 
 
Allgemein 
[ ba. ] ] wird in n Teilintervalle [ ]  mit ,  und n 
gerade zerlegt. In jeweils für zwei aufeinanderfolgenden Teilintervalle wird die 
Keplersche Fassregel angewendet und das Integral durch Summation der Ergebnisse 
näherungsweise berechnet: 

[ ] [ nn xxxxxx .,.,. 12110 −L ax =0 bxn =

 

( ) ( ) ( ) ([ ]nnn

b

a
yyyyyyyyyhdxxfF +++++++++≈= −−∫ 12432210 444

3
L )  

 

Simpsonregel:  ( ) ( )nnn

b

a
yyyyyyyyhdxxfF ++++++++≈= −−∫ 1243210 422424

3
L  

 
 
Beispiel 

Berechnen des Integral ∫ +
1

0 1 x
dx  durch quadratische Interpolation (s.o.) ⇒ , 

, , 

6=n

0=a 1=b
6
1=h : 

 

Simsonregel: ≈





 ++++++≈

+∫ 2
1

11
24

5
6

3
8

2
3

7
241

18
1

1
1

0 x
dx  0.6932 (TR) 

 
Programm Simpson.bas : 0.6931682 (einfache Genauigkeit) 
 
Beide Ergebnisse sind besser als die der Trapezregel. Das mit der Simpsonregel 
entwickelte Interpolationspolynom liefert eine bessere Annäherung der 
Ausgangsfunktion. 

Simpson.bas 
REM Berechnung eines bestimmten Integrals 
REM mittels Simpsonregel 
REM M. Meiler, 23.01.1998 
 
REM Hauptprogramm 
CLS : GOSUB eingabe: GOSUB simpson: GOSUB ausgabe 
END 
 
 
REM Eingabe der obere und unteren Grenze 
REM und der  Stuetzstellen 
eingabe: 
DO 
  INPUT "Untere Integrationsgrenze a= ", a! 
  INPUT "Obere Integrationsgrenze b(b>a)= ", b! 
LOOP UNTIL (b! - a!) > 0 
 
DO 
  INPUT "Anzahl der Stuetzstellen (s>1, ungerade) "; s% 
LOOP UNTIL s% >= 2 AND s% MOD 2 = 1 
 
n% = s% - 1 
h! = (b! - a!) / n% 
DIM y!(n%) 
 
FOR i% = 0 TO n% 
  LOCATE , 5 
  PRINT "x("; i%; ")="; a! + i% * h!; 
  LOCATE , 30 
  PRINT "y("; i%; ")="; 
  INPUT " ", y!(i%) 
NEXT i% 
RETURN 
 
 
REM Berechnung des Integrals mittels der Simpsonregel 
simpson: 
y! = y!(0) + y!(n%) 
 
FOR j% = 1 TO n% - 1 STEP 2 
  y! = y! + 4 * y!(j%) 
NEXT j% 
 
FOR j% = 2 TO n% - 2 STEP 2 
  y! = y! + 2 * y!(j%) 
NEXT j% 
 
y! = y! * h / 3 
RETURN 
 
 
REM Ausgabe des Ergebnisses 
ausgabe: 
PRINT " Ergebnis: "; y! 
PRINT 
RETURN 

5.4 Polynomberechnung mittels Hornerschema 
Polynome spielen in der Numerik eine besondere Rolle. Im vorherigen Abschnitt 
verwendete man für die Berechnung von Flächenintegralen Interpolationspolynome, 
welche die Funktion unter dem Integral angenähert darstellen und deren Integral man 
leicht berechnen kann. Oft kennt man die zu betrachteten Funktion überhaupt nicht 
und hat nur einige Funktionswerte (Messwerte) zur Verfügung. Auch dann wird diese 
Funktion mittels der diskret gegebenen Funktionswerte durch Interpolationspolynome 
approximiert, in dem man Polynome durch die vorhandenen diskreten Punkte legt. 
 
 

5.4.1 Polynomberechnung 
Grundgedanke 
Durch geeignetes Ausklammern verhindert man das zeitaufwendige Berechnen der 
Potenzen in einem Polynom und spart damit Rechenzeit. 
Ausgangspunkt 

Polynom r-ten Grades ( ) 01
1

1 axaxaxaxP r
r

r
rr ++++= −

− L  
Beispiel 
Gegeben sei das Polynom 5. Grades , gesucht sei der 
Wert des Polynoms an der Stelle x

( ) 9352 245
5 −+−−= xxxxxP

( )35P0 = 3, also . Nun formt man das Polynom so 
um, dass keine Potenzen mehr vorhanden sind. 
 

( ) 9352 245
5 −+−−= xxxxxP ( )( )( )( ) 935012 −+−+−= xxxxx  

 
Diese Darstellungsweise gestattet eine sehr schnelle Berechnung, auch mit dem 
Taschenrechner. 
Das hierfür verwendete Rechenschema trägt den Namen Hornerschema, nach William 
Georg Horner (1786 - 1837), und sieht folgendermaβen aus: 
 

2 -1 0 -5 3 -9 
0 +           6 +         15 +         45 +       120 +       369 +            3

2 *      5 *     15 *     40 *    123 *    360 = P5(3) 

 
Die Probe zeigt, dass das Ergebnis korrekt ist. 
 
Allgemein 
 

( ) 01
1

1 axaxaxaxP r
r

r
rr ++++= −

− L ( )( ) 011 axaxaxa rr ++++= − LL  
 
Hornerschema für die Polynomwertbestimmung an der Stelle x0 
 

Informatik/Numerik (2002)  5-11/27 

ar ar-1 . . . a1 a0 
0 +      b0 x0   + . . . +    br-2 x0 +     br-1 x0 +           x0  

b0=ar *      b1       * . . . *    br-1 *      br = Pr(x0) 



Universität Leipzig  Institut für Informatik 
Dr. Monika Meiler 

mit b0 = ar, bi+1 = ar-i-1 + bi x0 für i = 1, ..., r und br = Pr (x0). 
 
Durch dieses Verfahren wird der Rechenaufwand wesentlich veringert: 
 
Polynomberechnung ohne Hornerschema: 

# Mult = ( )( )
2

211

1

++=∑
+

=

rri
r

i
 

# Add = r 

# Mult + # Add = 
2

25
2

223 22 ++=+++ rrrrr  

 
Polynomberechnung mit Hornerschema: 

# Mult = r 
# Add = r  
# Mult + # Add = 2r 

 
Horner1.bas 

REM Polynomwertbestimmung mittels Hornerschema 
REM M. Meiler, 21.01.2002 
 
DO 
  CLS 
  GOSUB polynomeingabe: PRINT 
  PRINT "Eingegebenes Polynom": GOSUB polynomausgabe: PRINT 
 
  DO 
    GOSUB argumenteingabe: PRINT: GOSUB horner1 
    GOSUB polynomwertausgabe: PRINT 
    PRINT "Berechnen des gleichen Polynoms "; 
    INPUT "mit einem neuen Argument (j/n)? ", Weiter$ 
  LOOP WHILE Weiter$ = "j" 
 
  INPUT "Berechnen eines neuen Polynoms(j/n)? ", Weiter$ 
LOOP WHILE Weiter$ = "j" 
END 
 
 
REM Eingabe des Grades und der Koeffizienten des Polynoms 
polynomeingabe: 
DO 
  INPUT "Grad des Polynoms(r > 0): ", r% 
LOOP UNTIL r% > 0 
 
REDIM a!(r%) 
 
PRINT "Koeffizienteneingabe" 
FOR i% = 0 TO r% 
  PRINT "a!("; i%; ")="; : INPUT " ", a!(i%) 
NEXT i% 
RETURN 
 
 
REM Eingabe des Arguments zur Berechnung des Polynomwertes 
argumenteingabe: 
INPUT "Argument x = ", x! 
RETURN 

REM Berechnung des Polynomwertes mittels Horner 
horner1: 
b! = 0 
 
FOR k% = r% TO 0 STEP -1 
  b! = a!(k%) + b! * x! ': PRINT "b("; k%; ")="; b! 
NEXT k% 
 
y! = b! 
RETURN 
 
 
REM Ausgabe des Polynoms in ueblicher Schreibweise 
polynomausgabe: 
FOR ii% = 0 TO r% 
  PRINT a!(ii%); "x^"; ii%;: IF ii% <> r% THEN PRINT "+ "; 
NEXT ii% 
PRINT 
RETURN 
 
 
REM Ausgabe des berechneten Polynomwertes 
polynomwertausgabe: 
PRINT "P("; x!; ")="; y! 
RETURN 

 
 

5.4.2 Berechnung der 1. Ableitung eines Polynoms 
 
Aus dem Hornerschema geht hervor, dass 

b0 = ar, bi+1 = ar-i-1 + bi  x0 für i = 1, ..., r und br = Pr (x0). 
 

Umgeformt ergibt sich 
ar = b0 und ar-i-1 = bi+1 - bi  x0. 
 

Setze j = r � i � 1, so ist 
ar = b0 und aj = br-j � br-j-1  x0 und wegen br = Pr (x0) folgt 

 
( ) 01

1
1

1
1 axaxaxaxaxP r

r
r

r
r

rr +++++= −
−

−
− L  

( ) ( ) ( ) ( 01021
2

012
1

0010 xbbxxbbxxbbxxbbxb rrrr
rrr

−−−
−− −+−++−+−+= L )  

     ---------- --------       ------------       ----------   --------  --------------- 

( ) ( ) ( ) ( ) rrr
rr bbxxxbxxxbxxxbxx +−+−++−+−= −−
−−

1020
2

10
1

00 L  
 ---------------------------- --------------------------- -------------------------- 

( )( ) rrr
rr bbxbxbxbxx +++++−= −−
−−

12
2

1
1

00 L  
 ---- 

⇒ ( ) ( )( ) ( )012
2

1
1

00 xPbxbxbxbxxxP rrr
rr

r +++++−= −−
−− L  

   --------------- 

⇒ ( ) ( ) ( )xp
xx

xPxP
r

rr =
−
−

0

0  mit  ( ) 12
2

1
1

0 −−
−− ++++= rr

rr
r bxbxbxbxp L

⇒  kann zur Berechnung der 1. Ableitung an der Stelle x = x( )xpr 0 genutzt 
werden: 

 
Differenzenquotient 
 

( ) ( ) ( ) ( 0
0

0

0

lim xP
xx

xPxP
dx

xdP
xx

′=
−
−=

→
)   mit   ( ) 12

2
1

1
0 −−

−− ++++=′ rr
rr bxbxbxbxP L

 
 
Erweitertes Hornerschema 
 
Beispiel 

( ) 9352 245
5 −+−−= xxxxxP , gesucht . ( )35P′

 
2 -1 0 -5 3 -9 
0 +           6 +         15 +         45 +       120 +       369 +            3

2 *      5 *     15 *     40 *    123 *    360 

0 +           6 +         33 +         144 +       552   + 3

2 *      11 *     48 *     184 *    675  

 
Probe  

( ) 9352 245
5 −+−−= xxxxxP , ,  ( ) 310410 34

5 +−−=′ xxxxP
( ) 36099458148635 =−+−−=P , . ( ) 67533010881035 =+−−=′P

 
Allgemein 
 
Erweitertes Hornerschema für die Polynomwertbestimmung und der 1. Ableitung an 
der Stelle x0 
 

ar ar-1 . . . a1 a0 
0 +      b0 x0   + . . . +    br-2 x0 +     br-1 x0 +           x0  

b0=ar *      b1       * . . . *    br-1 *      br = Pr(x0) 

0 +      c0 x0   + . . . +    cr-2 x0  + x0  

c0 = b0=ar *      c1       * . . . *    cr-1 = P'r(x0)  

 
mit c0 = b0 = ar, 
bi+1 = ar-i-1 + bi  x0 für i = 1, ..., r, 
ci+1 = bi+1 + ci  x0 für i = 1, ..., r-1, 
br = Pr (x0) und cr-1 = P'r (x0). 

 
Ein weiteres Beispiel: 

( ) 254 34
4 −+−= xxxxP , P4(4) = 706, P'4(4) = 785 

Horner2.bas 
REM Polynomwertbestimmung mittels 
REM erweiterten HORNER -Schema 
REM M. Meiler, 21.01.2002 
 
DO 
  CLS 
  GOSUB polynomeingabe: PRINT 
  PRINT "Eingegebenes Polynom": GOSUB polynomausgabe: PRINT 
  DO 
    GOSUB argumenteingabe: PRINT: GOSUB horner2 
    GOSUB polynomwertausgabe: PRINT 
 
    PRINT "Berechnen des gleichen Polynoms "; 
    INPUT "mit neuem Argument(j/n)? ", Weiter$ 
  LOOP WHILE Weiter$ = "j" 
  INPUT "Berechnen eines neuen Polynoms(j/n)? ", Weiter$ 
 
LOOP WHILE Weiter$ = "j" 
END 
 
 
REM Eingabe des Grads und der Koeffizienten des Polynoms 
polynomeingabe: 
DO 
  INPUT "Grad des Polynoms (r > 0): ", r% 
LOOP UNTIL r% > 0 
 
REDIM a!(r%) 
 
PRINT "Koeffizienteneingabe" 
FOR i% = 0 TO r% 
  PRINT "a!("; i%; ")="; : INPUT " ", a!(i%) 
NEXT i% 
RETURN 
 
 
REM Eingabe des Arguments zur Berechnung des Polynomwertes 
argumenteingabe: 
INPUT "Argument x = ", x! 
RETURN 
 
 
REM Berechnung des Polynomwertes und der ersten Ableitung 
REM mittels erweitertem Hornerschema 
horner2: 
b! = 0: c! = 0 
 
FOR k% = r% TO 0 STEP -1 
  b! = a!(k%) + b! * x! : 'PRINT "b("; k%; ")="; b!; 
  IF k% <> 0 THEN 
    c! = b! + c! * x! : 'PRINT "c("; k%; ")="; c! 
  END IF 
NEXT k% 
 
y! = b! : y1! = c! 
RETURN 
 
 
REM Ausgabe des Polynoms in ueblicher Schreibweise 
polynomausgabe: 

FOR ii% = 0 TO r% 
  PRINT a!(ii%); "x^"; ii%; 
  IF ii% <> r% THEN PRINT "+ "; 
NEXT ii% 
PRINT 
RETURN 
 
 
REM Ausgabe des berechneten Polynomwertes 
polynomwertausgabe: 
PRINT "P("; x!; ")="; y!: PRINT "P'("; x!; ")="; y1! 
RETURN 

 
 

5.4.3 Anwendungsbeispiel 
Bei der numerischen Integration (s. Abschnitt 5.3) können die Funktionswerte der 
Stützstellen mittels Hornerschema berechnet werden. Damit wird deren Eingabe 
überflüssig und Eingabefehler vermieden: 
 
PTrapez.bas 

REM Integration von Polynomen mittels Trapezregel 
REM Polynomwertbestimmung mittels HORNER-Schema 
REM M. Meiler, 20.01.2003 
 
REM Hauptprogramm 
CLS : GOSUB eingabe: GOSUB trapez: GOSUB ausgabe 
END 
 
 
REM Eingabe der obere und unteren Grenze 
REM und Berechnung der Stuetzstellen 
eingabe: 
GOSUB polynomeingabe: PRINT 
GOSUB grenzeneingabe: PRINT 
GOSUB stuetzstellen: PRINT 
RETURN 
 
 
REM Eingabe des Grades und der Koeffizienten des Polynoms 
polynomeingabe: 
DO 
  INPUT "Grad des Polynoms(r > 0): ", r% 
LOOP UNTIL r% > 0 
REDIM a!(r%) 
PRINT "Koeffizienteneingabe" 
FOR i% = 0 TO r% 
  PRINT "a!("; i%; ")="; : INPUT " ", a!(i%) 
NEXT i% 
RETURN 
 
 
REM Eingabe der obere und unteren Grenze 
grenzeneingabe: 
DO 
  INPUT "Untere Integrationsgrenze a= ", a! 
  INPUT "Obere Integrationsgrenze b(b>a)= ", b! 
LOOP UNTIL (b! - a!) > 0 
RETURN 

REM Berechnung der Stuetzstellen mittels HORNER 
stuetzstellen: 
DO 
  INPUT "Anzahl der Stuetzstellen (s > 1): ", s% 
LOOP UNTIL s% >= 2 
 
n% = s% - 1 
h! = (b! - a!) / n% 
REDIM y!(n%) 
 
FOR i% = 0 TO n% 
  x! = a! + i% * h! 
  GOSUB horner1: y!(i%) = y! 
NEXT i% 
RETURN 
 
 
REM Berechnung des Polynomwertes mittels Horner 
horner1: 
b! = 0 
FOR k% = r% TO 0 STEP -1 
  b! = a!(k%) + b! * x! 
NEXT k% 
y! = b! 
RETURN 
 
 
REM Berechnung des Integrals mittels Trapezregel 
trapez: 
y! = y!(0) + y!(n%) 
 
FOR j% = 1 TO n% - 1 
  y! = y! + 2 * y!(j%) 
NEXT j% 
 
y! = y! * h! / 2 
RETURN 
 
 
REM Ausgabe des Ergebnisses 
ausgabe: 
PRINT "Ergebnis: "; y! 
PRINT 
RETURN 

5.5 Iterationsverfahren 
Iterationsverfahren berechnen ausgehend von einer Anfangsnäherungen eine bessere 
Näherung. Das wird solange wiederholt (iteriert), bis eine gewünschte Genauigkeit 
erreicht wurde. Sie haben mit der Entwicklung der Computertechnik durch 
Rechengeschwindigkeit und Rechengenauigkeit an Bedeutung gewonnen. 
 
 

5.5.1 Newtonverfahren (Tangentenverfahren) 
Das nach Isaac Newton (1643-1727) benannte Verfahren dient der 
Nullstellenbestimmung einer gegebenen Funktion. 
 
Grundgedanke 
Ausgehend von einer bereits vorhandenen Näherung  wird im Punkt 

 an die Funktion  eine Tangente t  gelegt. Der Schnittpunkt 
dieser Tangente mit der Abzisse liefert eine neue Näherung . Der Vorgang wird so 
oft wiederholt, bis man eine geeignete Näherung einer Nullstelle 

0x
( )( 000 , xfxP = ) ( )xf ( )x

1x
x  gefunden hat. 

 
 

_
xx0 x1

x

P0

t(x)
f(x)

y 

 
 
Ausgangspunkt 
Eine stetig differenzierbare Funktion  und eine Näherung . ( )xf 0x
 
Berechnung der Tangente  in P( )xt 0 an : ( )xf

Die Tangente t  an die Funktion  durch den Punkt , wobei 
, wird mittels Punktrichtungsgleichung berechnet. Sei  die 

Steigung von  im Punkt P

( )x

( )xf

( )xf ( 000 , yxP =

0 fy ′=′
)

) )( 00 xfy = ( 0x
0, so ist 

( )000 xxyyy −′=−  
und somit 

( ) ( )000 xxyyyxt −′+==  
Tangente in P0 an . ( )xf

Berechnung der Nullstelle  von : 1x ( )xt

( ) 01 =xt  ⇒  ⇒ ( ) 00100 =−′+ xxyy
0

0
01 y

yxx
′

−=−  ⇒ 
0

0
01 y

yxx
′

−=  

⇒ 

Newtonverfahren: ( )
( )0

0
01 xf

xfxxx
′

−=≈  

 
Allgemein 
Ausgehend von einer Näherung  wird eine Folge von Näherungen  
berechnet, die gegen 

0x nxxx ,,, 21 L

x  konvergiert.  
 

 

t2(x) 

t1(x)

x2

x3

x0 x1
x

P0 

t0(x)

f(x)

y 

 
 

5.5.2 Anwendungsbeispiel 1 
Berechnung der k-ten Wurzel aus a mit , ; ,  Ν∈k 1>k R∈a 0>a

k ax =  ⇒  ( ) 0=−= axxf k

⇒ Das Problem reduziert sich auf die Nullstellenbestimmung von 
( ) axxf k −=  mit  ( ) 1−⋅=′ kxkxf

mittels Newtonverfahren 1
0

0
01 −⋅

−−= k

k

xk
axxx . 

 
Speziell für ,  ⇒   und dem Anfangwert  
ergibt sich mittels Taschenrechner aus dem Newtonverfahren: 

4=a 2=k ( ) axxf −= 2 ( ) xxf ⋅=′ 2 40 =x

 

i ix  axy ii −= 2  ii xy ⋅=′ 2  
i

i
ii y

yxx
′

−=+1  

0 4 12 8 4 - 12/8 = 2.5 
1 2.5 2.25 5 2.5 �2.25/5 = 2.05 
2 2.05 0.203 4.1 2.05 � 0.203/4.1 = 2 
3 2 0 (Abbruch)   

 
Das Verfahren bricht für  x3 = 2  ab und konvergiert damit gegen den korrekten 
Wurzelwert. 

Newton.bas 
 

REM Newtonverfahren (Tangentenverfahren) 
REM Berechnung der k. Wurzel aus a 
REM M. Meiler, 26.01.2001 
 
' k%   ... Wurzelexponent 
' a#   ... Radikant 
' EPS# ... Genauigkeit fuer Abbruchbedingung 
' x0#  ... Naeherung, anfangs a# 
' y0#  ... Funktionswert an der Stelle x0# 
' y1#  ... 1. Ableitung an der Stelle x0# 
 
 
CLS 
DO 
  GOSUB eingabe: GOSUB newton: GOSUB ausgabe 
  INPUT "Weiter (j/n)? ", Weiter$ 
LOOP WHILE Weiter$ = "j" 
END 
 
 
eingabe: 
DO 
  PRINT "Wurzelberechnug  -  k. Wurzel aus a  - (k>1, a>0)": PRINT 
  INPUT "k = ", k% 
  INPUT "a = ", a# 
LOOP UNTIL k% > 1 AND a# > 0 
PRINT 
DO 
  PRINT "Genauigkeit (0<EPS<1)": PRINT 
  INPUT "EPS = ", EPS# 
LOOP UNTIL 0 < EPS# AND EPS# < 1 
RETURN 
 
 
newton: 
x0# = a#: y0# = x0# ^ k% - a#: y1# = k% * x0# ^ (k% - 1) 
 
DO 
  'Testausgabe 
  'LOCATE , 10: PRINT x0# 
 
  'Testausgabe 
  'LOCATE , 10: PRINT y0#; : LOCATE , 40: PRINT y1#: PRINT 
 
  IF ABS(y0#) < EPS# THEN EXIT DO 
 
  x0# = x0# - y0# / y1#                           ' Newton 
  y0# = x0# ^ k% - a#: y1# = k% * x0# ^ (k% - 1) 
 
LOOP 
RETURN 
 
 
ausgabe: 
PRINT "Die "; k%; ". Wurzel von "; a#; " ist "; x0#; " !" 
RETURN 

Protokoll zum Programm Wurzel.bas mit den Eingabewerten: 
a# = 4, k% = 2 , EPS# = .0001. 

 
 
k% a# ESP# x0# y0# y1# Bemerkungen 

2 4 .0001    Eingabewerte 
   4 12 8 Startwerte 
   2.5 2.25 5 1. Schleifendurchlauf 
   2.05 0.202 5 4.1 2. Schleifendurchlauf 
   2.00061 0.002 44 4.001 22 3. Schleifendurchlauf 
   2.000 000 1 0.000 000 37 4.000 000 4 4. Schleifendurchlauf 

Abbruch 
 
 

5.5.3 Anwendungsbeispiel 2 
Nullstellenberechnung von Polynomen. 
Gegeben:  ( ) ( ) 01

1
1 axaxaxaxPxf r

r
r

rr ++++== −
− L

Näherung x0 

Gesucht: Näherung ( )
( )0

0
01 xf

xfx
′

−=x , 

wobei  und  mittels Hornerschema berechnet werden. ( 0xf ) )

) )

( 0xf ′
 
Programmschritte: 
1. Eingabe des Polynoms (Unterprogramm eingabe im Programm Horner2.bas). 
2. Eingabe der Genauigkeit EPS der Berechnung. 
3. Berechnung einer Nullstellennäherung. 

(a) Eingabe der Anfangsnäherung x0. 
(b) Berechnung des Funktionswertes  und dessen Ableitung  mittels 

erweitertem Hornerschema (Unterprogramm horner2 im Programm 
Horner2.bas). 

( 0xf ( 0xf ′

(c) Ist ( ) EPSyf ≥0 : 
Berechnung einer neuen Näherung x1 mittels Newtonverfahren. 
Setze x0 = x1 und weiter mit 3 (b). 

(d) Näherung x1 für eine Nullstelle wurde gefunden. 
4. Soll eine neue Nullstellennäherung berechnet werden, weiter bei 3.  
5. Soll die Genauigkeit verändert werden, weiter bei 2.  
6. Soll ein neues Polynom berechnet werden, weiter bei 1. 
7. Fertig 
 
 
PNewton.bas 

REM Newtonverfahren (Tangentenverfahren) 
REM Nullstellenberechnung in Polynomen 
REM M. Meiler, 26.01.2001 
 
' EPS#    ... Genauigkeit fuer Abbruchbedingung 
' r%      ... Grad des Polynoms 
' a#( i%) ... i. Koeffizient des Polynoms 

' a!      ... Startwert 
' x0#     ... Naeherung, anfangs a! 
' y0#     ... Funktionswert an der Stelle x0# 
' y1#     ... 1. Ableitung an der Stelle x0# 
' x#      ... Argument fuer Polynomberechnung nach Horner 
' b#      ... Funktionswert an der Stelle x# nach Horner 
' c#      ... 1. Ableitung an der Stelle x# nach Horner 
 
 
DO 
  CLS 
  GOSUB polynomeingabe 
  DO 
    GOSUB epseingabe 
    DO 
      GOSUB startwerteingabe 
      GOSUB newton 
      GOSUB ausgabe 
      INPUT "Neuer Startnaeherungswert (j/n)? ", Weiter$ 
    LOOP WHILE Weiter$ = "j" 
    INPUT "Neue Genauigkeit (j/n)? ", Weiter$ 
  LOOP WHILE Weiter$ = "j" 
  INPUT "Neues Polynom (j/n)? ", Weiter$ 
LOOP WHILE Weiter$ = "j" 
END 
 
 
polynomeingabe: 
DO 
  INPUT "Grad des Polynoms: ", r% 
LOOP UNTIL r% > 0 
REDIM a#(r% + 1) 
PRINT "Koeffizienteneingabe" 
FOR i% = 0 TO r% 
  PRINT "a("; i%; ")="; : INPUT " ", a#(i%) 
NEXT i% 
RETURN 
 
 
epseingabe: 
DO 
  PRINT "Genauigkeit (0<EPS<1)": PRINT 
  INPUT "EPS = ", EPS# 
LOOP UNTIL 0 < EPS# AND EPS# < 1 
RETURN 
 
 
startwerteingabe: 
INPUT "Startwert a= ", a! 
x0# = a! 
RETURN 
 
 
newton: 
x# = x0#: GOSUB horner: y0# = b#: y1# = c# 
DO 
  ' Testausgabe 
  ' LOCATE , 10: PRINT x0# 
  ' Testausgabe 
  ' LOCATE , 10: PRINT y0#; : LOCATE , 40: PRINT y1#: PRINT 

  IF ABS(y0#) < EPS# THEN EXIT DO 
 
  x0# = x0# - y0# / y1#                            ' Newton 
  x# = x0#: GOSUB horner: y0# = b#: y1# = c# 
LOOP 
RETURN 
 
 
horner: 
b# = 0: c# = 0 
FOR k% = r% TO 0 STEP -1 
  b# = a#(k%) + b# * x# 
  IF k% <> 0 THEN c# = b# + c# * x# 
  ' PRINT "b="; b#; "c="; c# 
NEXT k% 
RETURN 
 
 
ausgabe: 
PRINT "Das Newtonverfahren auf das Polynom mit den Koeffizienten" 
FOR ii% = 0 TO r% 
  PRINT "a("; ii%; ")="; a#(ii%) 
NEXT ii% 
PRINT "und den Startwert "; a!; "angewand," 
PRINT "liefert als Nullstelle den Wert "; x0#; "!" 
PRINT 
RETURN 

 
 

5.5.4 Regula falsi (Sekantenverfahren) 
Auch hier sollen die Nullstellen einer Funktion  berechnet werden. ( )xf
 
Grundgedanke 
Ausgehend von zwei bereits vorhandenen Näherung ,  mit  und 

 wird durch die Punkte  und  eine Sekante 
 gelegt. Der Schnittpunkt dieser Sekante mit der Abzisse liefert eine neue 

Näherung . Der Vorgang wird so oft wiederholt, bis man eine geeignete Näherung 
einer Nullstelle 

0x

1P =
1x

1,x
( ) 00 <xf

))( ) 01 >xf
( )xs

( )( )000 , xfxP = (( 1xf

2x
x  gefunden hat. 

 
 

P1

_
xx0 

x2

x

P0

s(x)

f(x)

y 

x1

Ausgangspunkt 
Funktion  und zwei Nullstellennäherung  und  mit  und . ( )xf 0x 1x ( ) 00 <xf ( ) 01 >xf
 
Berechnung der Sekante  durch P( )xs 0 und P1: 

Die Sekante  durch die Punkte  und , wobei 
, , wird mittels Zweipunktegleichung berechnet. 

( )xs
f=

( 000 , yxP = ) )

)

( 111 , yxP =
( )00 xfy = ( )11 xy

( )( ) ( )( ) 0010010 =−−−−− yyxxxxyy  
und somit 

( ) ( )(
( )01

010
0 xx

yyxxyyxs
−

−−+==  

Sekante durch P0 und P1. 
 
Berechnung der Nullstelle  von : 2x ( )xs

( ) 02 =xs  ⇒ 
( )( )

( ) 0
01

0102
0 =

−
−−+

xx
yyxxy  ⇒ ( )

( )01

01
002 yy

xxyxx
−
−−=−  

⇒ 

Regula falsi: ( )
( )01

01
002 yy

xxyxxx
−
−−=≈  

 
Allgemein 
Ausgehend von zwei Näherung ,  wird eine Folge von Näherungen  
berechnet, die gegen 

0x 1x nxxx ,,, 21 L

x  konvergiert.  
 

 

_
x

P2

P1

x0 

x2

x

P0

s1(x)

f(x)

y 

x1

 

x3

s2(x)

 
Der Vorteil dieses Verfahrens gegenüber des Newtonverfahrens ist, dass man auf die 
Berechnung der ersten Ableitung verzichten kann. Der Nachteil besteht in der 
langsameren Konvergenz. 
 
 

5.5.5 Anwendungsbeispiel 3 
Berechnung der k-ten Wurzel aus a mit , ; ,  analog s. o. Ν∈k 1>k R∈a 0>a

k ax =  ⇒  ( ) 0=−= axxf k

⇒ Das Problem reduziert sich wiederum auf die Nullstellenbestimmung von 
( ) axxf k −=  

 
Speziell für ,  ⇒  und den Anfangwerten ,  
ergibt sich mittels Taschenrechner und Regula falsi: 

4=a 2=k ( ) axxf −= 2 10 =x 41 =x

 

i ix  1+ix axy ii −= 2  axy ii −= ++
2

11
ii

ii
iii yy

xxyxx
−
−−=

+

+
+

1

1
2  axy ii −= ++

2
22

0 1 4 -3 12 1.6 -1.44 
1 1.6 4 -1.44 12 1.8571429 -0.55102 
2 1.8571429 4 -0.55102 12 1.951295 -0.19244 
3 1.951295 4 -0.19244 12 1.9836308 -0.0652087 
4 1.9836308 4 -0.0652087 12 1.9943517 -0.0225613 
 
Das Verfahren konvergiert gegen den korrekten Wurzelwert. 
 
 
Regula.bas 

REM Regula falsi (Sekanten-Verfahren) 
REM Berechnung der k. Wurzel aus a 
REM M. Meiler, 26.01.2001 
 
' k%   ... Wurzelexponent 
' a#   ... Radikant 
' EPS# ... Genauigkeit fuer Abbruchbedingung 
' x0#  ... 1. Naeherungswert 
' x1#  ... 2. Naeherungswert 
' x2#  ... neuer Naeherungswert 
' y0#  ... Funktionswert an der Stelle x0# 
' y1#  ... Funktionswert an der Stelle x1# 
' y2#  ... Funktionswert an der Stelle x2# 
 
CLS 
DO 
  GOSUB eingabe: GOSUB regula: GOSUB ausgabe 
  INPUT "Weiter (j/n)? ", Weiter$ 
LOOP WHILE Weiter$ = "j" 
END 
 
 
eingabe: 
DO 
  PRINT "Wurzelberechnug  -  k. Wurzel aus a  - (k>1, a>0)" 
  INPUT "k = ", k% 
  INPUT "a = ", a# 
LOOP UNTIL k% > 1 AND a# > 0 
PRINT 
DO 
  PRINT "Genauigkeit (0<EPS<1)": PRINT 
  INPUT "EPS = ", EPS# 
LOOP UNTIL 0 < EPS# AND EPS# < 1 
DO 
  PRINT "Startwerte x0 und x1 mit x0 < k. Wurzel aus a < x1" 
  INPUT "x0 = ", x0# 
  INPUT "x1 = ", x1# 

  y0# = x0# ^ k% - a#: y1# = x1# ^ k% - a# 
LOOP UNTIL y0# < 0 AND y1# > 0 
RETURN 
 
 
regula: 
DO 
  'Testausgabe 
  LOCATE , 10: PRINT x0#; : LOCATE , 40: PRINT x1# 
  'Testausgabe 
  LOCATE , 10: PRINT y0#; : LOCATE , 40: PRINT y1#: PRINT 
 
                                           ' Regula falsi 
  x2# = x0# - y0# * (x1# - x0#) / (y1# - y0#) 
  y2# = x2# ^ k% - a# 
  
  IF ABS(y2#) < EPS# THEN EXIT DO 
 
  IF SGN(y1#) = SGN(y2#) THEN 
    x1# = x2#: y1# = y2# 
  ELSE 
    x0# = x2#: y0# = y2# 
  END IF 
LOOP 
RETURN 
 
 
ausgabe: 
PRINT "Die "; k%; ". Wurzel von "; a#; " ist "; x2#; " !" 
RETURN 

 
 

5.6 Zusammenfassung 
 

Eingabefehler lassen sich in der Regel nur durch die Verbesserung der 
Messtechniken vermeiden. Die Toleranz, in der diese Fehler liegen, entscheiden 
über die zu verwendete Rechengenauigkeit und Genauigkeit der Verfahren. 

• 

• 

• 

 
Mathematisch exakte Verfahren können durch Rechenfehler zu unexakten 
Ergebnissen führen. Rechenfehler können durch die Verwendung anderer 
Datentypen minimiert aber nicht beseitigt werden. Solche Fehler sind auch bei 
hochentwickelter Computertechnik nicht auszuschließen. 
Gauss.bas. Das Programm zum Gaußverfahren liefert in der Regel keine 
Dreiecksmatrix, obwohl eine solche mathematisch zu erwarten ist. 

 
Näherungsverfahren sind mathematisch unexakte Verfahren. Sie ersetzen 
komplizierte Verfahren durch einfachere und liefern nur Näherungen der Lösung. 
Verfahrensfehler können durch die Verwendung anderer Datentypen in der 
Regel. nicht verbessert werden. 
Trapez.bas, Simpson.bas. Bei der Integration durch Polynominterpolation wurden 
die Funktion durch Polynome ersetzt, welche sich rechentechnisch besser 
integrieren lassen. Eine Verbesserung der Ergebnisse konnte durch mehr 
Stützstellen oder einen höheren Grad des Interpolationspolynoms, also der 
Verbesserung des verwendeten Verfahrens, erreicht werden. 

Wurzel.bas, Ragula.bas berechnen Wurzeln als Nullstellen einer Funktion mit 
unterschiedlichen Näherungsverfahren. Beide Verfahren benötigen bereits 
Näherungslösungen. Das Newtonverfahren konvergiert sehr schnell, benötigt aber 
die erste Ableitung der betrachteten Funktion. Wendet man die Verfahren auf 
Funktionen mit mehreren Nullstellen an, so findet man u. U. nicht alle Nullstellen 
der Funktion. Die Anfangsnäherung ist für das Ergebnis ausschlaggebend. 

 
Durch die vorhandenen Datentypen und Rechenoperationen sind oft Grenzen für 
die Berechnung gesetzt, Ergebnisse sind deshalb stets bzgl. ihrer Korrektheit 
abzuschätzen. 

• 

• 

Fak.bas. Die Fakultätsberechnung lieferte bei ganzen Zahlen bis 7!, bei langen 
ganzen Zahlen und rationalen Zahlen einfacher Genauigkeit bis 12! und bei 
rationalen Zahlen doppelter Genauigkeit bis 22! exakte Ergebnisse. Bei gröβeren 
Zahlen wurde die Berechnung durch Überlauf abgebrochen bzw. durch Runden 
falsch. 

 
Mathematisch elegante Verfahren sind rechentechnisch nicht immer die 
günstigsten. Rechentechnisch optimale Verfahren verbrauchen wenig Rechenzeit. 
Eine Rechenzeitminimierung kann man durch Minimierung der Anzahl der 
verwendeten Operationen erreichen. 
Horner1.bas. Bei der Polynomwertberechnung eines Polynoms vom Grad r ohne 

Hornerschema benötigt man 
2

252 ++ rr  Additionen und Multiplikationen, das 

sind zum Beispiel bei r = 4 insgesamt 19 Operationen. Durch die Verwendung des 
Hornerschemas konnte dieser Rechenaufwand auf 2 r verringert werden, bei r = 4 
sind das nur 8, also weniger als die Hälfte. 

 
Die modulare Programmierung gestatten die mehrfache Verwendung von schon 
bereits ausgetesteten Programmteilen, hier Unterprogrammen. In sich 
abgeschlossene Programmteile werden in einem Unterprogramm 
zusammengefasst. Diese können getrennt vom Hauptprogramm ausgetestet, 
jederzeit wieder aufgerufen und leicht durch andere ausgetauscht werden. 

• 

Horner1.bas, Horner2.bas, Tabelle.bas, PTrapez.bas, PIntegr.bas, 
PNewton.bas. Die Module zum Eingeben eines Polynoms, zur Berechnung eines 
Polynomwertes bzw. eines Polynomwertes und dessen Ableitung an einer 
gegebenen Stelle mittels Hornerschema können bei der Lösung verschiedener 
Probleme wiederverwendet werden, wie zum Beispiel beim Tabellieren eines 
Polynoms, bei der Integralberechnung eines Polynoms zum Berechnen der 
Funktionswerte an den Stützstellen oder wie im letzten Abschnitt, bei der 
Nullstellenbestimmung eines Polynoms zur Berechnung des Polynomwertes und 
deren Ableitung von einer Näherung. 

